3 - Problemi e modelli di ottimizzazione

Mauro Passacantando

Dipartimento di Scienze Economico-Aziendali e Diritto per l'Economia
Università degli Studi di Milano-Bicocca
mauro.passacantando@unimib.it

Corso di Dinamica dei Sistemi Aziendali
Laurea Magistrale in Scienze Economico-Aziendali
Università degli Studi di Milano-Bicocca

Problema 1 - descrizione

Un'azienda produttrice di collettori rotanti ha appena ricevuto un ordine di acquisto. L'azienda produce 3 diversi modelli di collettori. Sono stati ordinati 3000 collettori del modello 1, 2000 del modello 2 e 900 del modello 3.
Per la produzione di ogni collettore sono necessarie due fasi di lavorazione. Le ore necessarie per effettuare le due fasi di lavorazione per ogni collettore di ogni modello sono indicate in tabella:

	Modello 1	Modello 2	Modello 3
Fase 1 (ore/unità)	2	1.5	3
Fase 2 (ore/unità)	1	2	1

L'azienda ha a disposizione 10000 ore per la fase 1 di lavorazione e 5000 ore per la fase 2.
Il costo totale di produzione per l'azienda è di 50€ per ogni collettore del modello 1, di 83€ per ogni collettore del modello 2 e di 130€ per ogni collettore del modello 3.
Per soddisfare l'ordine ricevuto, l'azienda ha anche la possibilità di comprare i collettori da un'azienda concorrente ai seguenti prezzi: 61€ per un collettore del modello 1, 97€ per uno del modello 2 e 145€ per uno del modello 3.

L'azienda deve decidere quanti collettori produrre di ogni modello e quanti acquistarne dall'azienda concorrente in modo da soddisfare l'ordine ricevuto, con l'obiettivo di minimizzare il costo complessivo.

Problema 1 - formulazione

Variabili decisionali:

$x_1 =$ numero di collettori del modello 1 realizzati "in casa"
$x_2 =$ numero di collettori del modello 2 realizzati "in casa"
$x_3 =$ numero di collettori del modello 3 realizzati "in casa"
$y_1 =$ numero di collettori del modello 1 comprati dall'azienda concorrente
$y_2 =$ numero di collettori del modello 2 comprati dall'azienda concorrente
$y_3 =$ numero di collettori del modello 3 comprati dall'azienda concorrente

Problema 1 - formulazione

Variabili decisionali:

x_1 = numero di collettori del modello 1 realizzati "in casa"
x_2 = numero di collettori del modello 2 realizzati "in casa"
x_3 = numero di collettori del modello 3 realizzati "in casa"
y_1 = numero di collettori del modello 1 comprati dall'azienda concorrente
y_2 = numero di collettori del modello 2 comprati dall'azienda concorrente
y_3 = numero di collettori del modello 3 comprati dall'azienda concorrente

Funzione obiettivo: $50x_1 + 83x_2 + 130x_3 + 61y_1 + 97y_2 + 145y_3$ (da minimizzare)

Problema 1 - formulazione

Variabili decisionali:

x_1 = numero di collettori del modello 1 realizzati "in casa"
x_2 = numero di collettori del modello 2 realizzati "in casa"
x_3 = numero di collettori del modello 3 realizzati "in casa"
y_1 = numero di collettori del modello 1 comprati dall'azienda concorrente
y_2 = numero di collettori del modello 2 comprati dall'azienda concorrente
y_3 = numero di collettori del modello 3 comprati dall'azienda concorrente

Funzione obiettivo: $50x_1 + 83x_2 + 130x_3 + 61y_1 + 97y_2 + 145y_3$ (da minimizzare)

Vincoli da rispettare:

$2x_1 + 1.5x_2 + 3x_3 \leq 10000$ (ore di lavorazione fase 1)

Problema 1 - formulazione

Variabili decisionali:

x_1 = numero di collettori del modello 1 realizzati "in casa"
x_2 = numero di collettori del modello 2 realizzati "in casa"
x_3 = numero di collettori del modello 3 realizzati "in casa"
y_1 = numero di collettori del modello 1 comprati dall'azienda concorrente
y_2 = numero di collettori del modello 2 comprati dall'azienda concorrente
y_3 = numero di collettori del modello 3 comprati dall'azienda concorrente

Funzione obiettivo: $50x_1 + 83x_2 + 130x_3 + 61y_1 + 97y_2 + 145y_3$ (da minimizzare)

Vincoli da rispettare:

$2x_1 + 1.5x_2 + 3x_3 \leq 10000$ (ore di lavorazione fase 1)
$x_1 + 2x_2 + x_3 \leq 5000$ (ore di lavorazione fase 2)

Problema 1 - formulazione

Variabili decisionali:

x_1 = numero di collettori del modello 1 realizzati "in casa"
x_2 = numero di collettori del modello 2 realizzati "in casa"
x_3 = numero di collettori del modello 3 realizzati "in casa"
y_1 = numero di collettori del modello 1 comprati dall'azienda concorrente
y_2 = numero di collettori del modello 2 comprati dall'azienda concorrente
y_3 = numero di collettori del modello 3 comprati dall'azienda concorrente

Funzione obiettivo: $50x_1 + 83x_2 + 130x_3 + 61y_1 + 97y_2 + 145y_3$ (da minimizzare)

Vincoli da rispettare:

$2x_1 + 1.5x_2 + 3x_3 \leq 10000$ (ore di lavorazione fase 1)
$x_1 + 2x_2 + x_3 \leq 5000$ (ore di lavorazione fase 2)
$x_1 + y_1 = 3000$ (domanda modello 1)

Problema 1 - formulazione

Variabili decisionali:

x_1 = numero di collettori del modello 1 realizzati "in casa"
x_2 = numero di collettori del modello 2 realizzati "in casa"
x_3 = numero di collettori del modello 3 realizzati "in casa"
y_1 = numero di collettori del modello 1 comprati dall'azienda concorrente
y_2 = numero di collettori del modello 2 comprati dall'azienda concorrente
y_3 = numero di collettori del modello 3 comprati dall'azienda concorrente

Funzione obiettivo: $50x_1 + 83x_2 + 130x_3 + 61y_1 + 97y_2 + 145y_3$ (da minimizzare)

Vincoli da rispettare:

$2x_1 + 1.5x_2 + 3x_3 \leq 10000$ (ore di lavorazione fase 1)
$x_1 + 2x_2 + x_3 \leq 5000$ (ore di lavorazione fase 2)
$x_1 + y_1 = 3000$ (domanda modello 1)
$x_2 + y_2 = 2000$ (domanda modello 2)

Problema 1 - formulazione

Variabili decisionali:

x_1 = numero di collettori del modello 1 realizzati "in casa"
x_2 = numero di collettori del modello 2 realizzati "in casa"
x_3 = numero di collettori del modello 3 realizzati "in casa"
y_1 = numero di collettori del modello 1 comprati dall'azienda concorrente
y_2 = numero di collettori del modello 2 comprati dall'azienda concorrente
y_3 = numero di collettori del modello 3 comprati dall'azienda concorrente

Funzione obiettivo: $50x_1 + 83x_2 + 130x_3 + 61y_1 + 97y_2 + 145y_3$ (da minimizzare)

Vincoli da rispettare:

$2x_1 + 1.5x_2 + 3x_3 \leq 10000$ (ore di lavorazione fase 1)
$x_1 + 2x_2 + x_3 \leq 5000$ (ore di lavorazione fase 2)
$x_1 + y_1 = 3000$ (domanda modello 1)
$x_2 + y_2 = 2000$ (domanda modello 2)
$x_3 + y_3 = 900$ (domanda modello 3)

Problema 1 - formulazione

Variabili decisionali:

x_1 = numero di collettori del modello 1 realizzati "in casa"
x_2 = numero di collettori del modello 2 realizzati "in casa"
x_3 = numero di collettori del modello 3 realizzati "in casa"
y_1 = numero di collettori del modello 1 comprati dall'azienda concorrente
y_2 = numero di collettori del modello 2 comprati dall'azienda concorrente
y_3 = numero di collettori del modello 3 comprati dall'azienda concorrente

Funzione obiettivo: $50x_1 + 83x_2 + 130x_3 + 61y_1 + 97y_2 + 145y_3$ (da minimizzare)

Vincoli da rispettare:

$2x_1 + 1.5x_2 + 3x_3 \leq 10000$ (ore di lavorazione fase 1)
$x_1 + 2x_2 + x_3 \leq 5000$ (ore di lavorazione fase 2)
$x_1 + y_1 = 3000$ (domanda modello 1)
$x_2 + y_2 = 2000$ (domanda modello 2)
$x_3 + y_3 = 900$ (domanda modello 3)
$x_1, x_2, x_3, y_1, y_2, y_3 \geq 0$
$x_1, x_2, x_3, y_1, y_2, y_3 \in \mathbb{Z}$ (variabili intere)

Problema 1 - formulazione

Modello matematico (di programmazione lineare intera)

$$
\begin{cases}
\min \ 50x_1 + 83x_2 + 130x_3 + 61y_1 + 97y_2 + 145y_3 \\
2x_1 + 1.5x_2 + 3x_3 \leq 10000 \\
x_1 + 2x_2 + x_3 \leq 5000 \\
x_1 + y_1 = 3000 \\
x_2 + y_2 = 2000 \\
x_3 + y_3 = 900 \\
x_1, x_2, x_3, y_1, y_2, y_3 \geq 0 \\
x_1, x_2, x_3, y_1, y_2, y_3 \in \mathbb{Z}
\end{cases}
$$

Problema 1 - formulazione AMPL

prod1.mod

```
#------------ parametri ---------------------
param n integer > 0; # numero di modelli da produrre
param domanda{i in 1..n} >= 0;
# domanda[i] = numero di collettori ordinati del modello i

param OreFase1{i in 1..n} >= 0;
# OreFase1[i] = numero di ore di lavorazione della fase 1
# necessarie per produrre un collettore del modello i

param cap1 >= 0;  # numero di ore disponibili per la fase 1 di lavorazione
param OreFase2{i in 1..n} >= 0;
# OreFase2[i] = numero di ore di lavorazione della fase 2
# necessarie per produrre un collettore del modello i

param cap2 >= 0;  # numero di ore disponibili per la fase 2 di lavorazione
param CostoProduzione{i in 1..n} >= 0;
# CostoProduzione[i] = costo per produrre un collettore del modello i

param CostoAcquisto{i in 1..n} >= 0;
# CostoAcquisto[i] = costo per acquistare
# dall'azienda concorrente un collettore del modello i
```

Problema 1 - formulazione AMPL

prod1.mod

```
#------------- variabili -------------------
var x{i in 1..n} >= 0 integer;
# x[i] = numero di collettori del modello i prodotti in casa

var y{i in 1..n} >= 0 integer;
# y[i] = numero di collettori del modello i acquistati
# dall'azienda concorrente

#------------- funzione obiettivo -------------------

minimize CostoTotale: sum{i in 1..n} CostoProduzione[i]*x[i]
+ sum{i in 1..n} CostoAcquisto[i]*y[i];

#------------- vincoli -------------------

s.t. VincoloFase1: sum{i in 1..n} OreFase1[i]*x[i] <= cap1;
s.t. VincoloFase2: sum{i in 1..n} OreFase2[i]*x[i] <= cap2;
s.t. VincoliDomanda{i in 1..n}: x[i] + y[i] = domanda[i];
```

Problema 1 - formulazione AMPL

prod1.dat

```
param n := 3;
param domanda :=
1 3000
2 2000
3 900;
param OreFase1 :=
1 2
2 1.5
3 3;
param cap1 := 10000 ;
param OreFase2 :=
1 1
2 2
3 1;
param cap2 := 5000 ;
param CostoProduzione :=
1 50
2 83
3 130;
param CostoAcquisto :=
1 61
2 97
3 145;
```

Problema 1 - formulazione AMPL

prod1.run

```
reset;
reset;

model prod1.mod; # carico il modello
data prod1.dat;  # carico i dati

option solver gurobi;
solve;

#------------ output -------------

printf"\nLa soluzione ottima è:\n\n";

printf"Produrre \n";
for{i in 1..n}{
 printf"%g collettori del modello %g \n",x[i],i;
}

printf"\ne acquistare \n";
for{i in 1..n}{
 printf"%g collettori del modello %g \n",y[i],i;
}
```

Problema 2 - descrizione

Un'azienda conserviera produce tonno all'olio, tonno al vapore e tonno agli aromi. Per la produzione di ogni tipo di tonno è necessario affittare una macchina avente i seguenti costi: 200€ a settimana per produrre tonno all'olio; 150€ a settimana per il tonno al vapore e 100€ a settimana per il tonno agli aromi. I tempi di lavorazione, le richieste di materia prima ed il ricavo per ogni prodotto sono riassunti nella seguente tabella:

prodotto	tempi di lavorazione (ore/scatola)	materia prima (kg/scatola)	ricavo (€ /scatola)
tonno all'olio	3	4	6
tonno al vapore	2	3	4
tonno agli aromi	6	4	7

Ogni settimana sono disponibili 150 ore di lavoro e 160 kg di tonno.

L'azienda deve stabilire il piano produttivo settimanale in modo da massimizzare il profitto complessivo.

Problema 2 - formulazione

Variabili decisionali:

x_1 = numero di scatole prodotte di tonno all'olio
x_2 = numero di scatole prodotte di tonno al vapore
x_3 = numero di scatole prodotte di tonno agli aromi

$$y_1 = \begin{cases} 1 & \text{se viene prodotto tonno all'olio,} \\ 0 & \text{altrimenti.} \end{cases}$$

$$y_2 = \begin{cases} 1 & \text{se viene prodotto tonno al vapore,} \\ 0 & \text{altrimenti.} \end{cases}$$

$$y_3 = \begin{cases} 1 & \text{se viene prodotto tonno agli aromi,} \\ 0 & \text{altrimenti.} \end{cases}$$

Problema 2 - formulazione

Variabili decisionali:

x_1 = numero di scatole prodotte di tonno all'olio
x_2 = numero di scatole prodotte di tonno al vapore
x_3 = numero di scatole prodotte di tonno agli aromi

$$y_1 = \begin{cases} 1 & \text{se viene prodotto tonno all'olio,} \\ 0 & \text{altrimenti.} \end{cases}$$

$$y_2 = \begin{cases} 1 & \text{se viene prodotto tonno al vapore,} \\ 0 & \text{altrimenti.} \end{cases}$$

$$y_3 = \begin{cases} 1 & \text{se viene prodotto tonno agli aromi,} \\ 0 & \text{altrimenti.} \end{cases}$$

Funzione obiettivo: $6x_1 + 4x_2 + 7x_3 - 200y_1 - 150y_2 - 100y_3$ (da massimizzare)

Problema 2 - formulazione

Variabili decisionali:

x_1 = numero di scatole prodotte di tonno all'olio
x_2 = numero di scatole prodotte di tonno al vapore
x_3 = numero di scatole prodotte di tonno agli aromi

$$y_1 = \begin{cases} 1 & \text{se viene prodotto tonno all'olio,} \\ 0 & \text{altrimenti.} \end{cases}$$

$$y_2 = \begin{cases} 1 & \text{se viene prodotto tonno al vapore,} \\ 0 & \text{altrimenti.} \end{cases}$$

$$y_3 = \begin{cases} 1 & \text{se viene prodotto tonno agli aromi,} \\ 0 & \text{altrimenti.} \end{cases}$$

Funzione obiettivo: $6x_1 + 4x_2 + 7x_3 - 200y_1 - 150y_2 - 100y_3$ (da massimizzare)

Vincoli da rispettare:

$3x_1 + 2x_2 + 6x_3 \leq 150$ (ore di lavoro)
$4x_1 + 3x_2 + 4x_3 \leq 160$ (kg di tonno)

Problema 2 - formulazione

Variabili decisionali:

x_1 = numero di scatole prodotte di tonno all'olio
x_2 = numero di scatole prodotte di tonno al vapore
x_3 = numero di scatole prodotte di tonno agli aromi

$$y_1 = \begin{cases} 1 & \text{se viene prodotto tonno all'olio,} \\ 0 & \text{altrimenti.} \end{cases}$$

$$y_2 = \begin{cases} 1 & \text{se viene prodotto tonno al vapore,} \\ 0 & \text{altrimenti.} \end{cases}$$

$$y_3 = \begin{cases} 1 & \text{se viene prodotto tonno agli aromi,} \\ 0 & \text{altrimenti.} \end{cases}$$

Funzione obiettivo: $6x_1 + 4x_2 + 7x_3 - 200y_1 - 150y_2 - 100y_3$ (da massimizzare)

Vincoli da rispettare:

$3x_1 + 2x_2 + 6x_3 \leq 150$ (ore di lavoro)
$4x_1 + 3x_2 + 4x_3 \leq 160$ (kg di tonno)
$x_1 \leq M\, y_1$ (legame tra x_1 e y_1)
[M è una costante sufficientemente grande, ad esempio $M = 100$.]
$x_2 \leq M\, y_2$ (legame tra x_2 e y_2)
$x_3 \leq M\, y_3$ (legame tra x_3 e y_3)
$x_1, x_2, x_3 \geq 0$
$x_1, x_2, x_3 \in \mathbb{Z}$ (variabili intere)
$y_1, y_2, y_3 \in \{0, 1\}$ (variabili binarie)

Problema 2 - formulazione

Modello matematico (di programmazione lineare intera)

$$
\begin{cases}
\max\ 6x_1 + 4x_2 + 7x_3 - 200y_1 - 150y_2 - 100y_3 \\
3x_1 + 2x_2 + 6x_3 \leq 150 \\
4x_1 + 3x_2 + 4x_3 \leq 160 \\
x_1 \leq M\, y_1 \\
x_2 \leq M\, y_2 \\
x_3 \leq M\, y_3 \\
x_1, x_2, x_3 \geq 0 \\
x_1, x_2, x_3 \in \mathbb{Z} \\
y_1, y_2, y_3 \in \{0, 1\}
\end{cases}
$$

Esercizio. Scrivere nella sintassi di AMPL il modello ed i dati relativi al problema precedente e trovare una soluzione ottima.

Problema 3 - descrizione

Un'azienda può produrre tre tipi di auto: utilitarie, berline, station-wagon. La quantità di acciaio necessaria, i tempi di lavorazione ed i profitti corrispondenti ad ogni tipo di auto sono sintetizzati nella seguente tabella:

tipi di auto	acciaio (tonnellate/auto)	lavoro (ore/auto)	profitto (euro/auto)
utilitarie	1.5	30	2000
berline	3	25	3000
station-wagon	5	40	4000

L'azienda ha a disposizione ogni mese 6000 tonnellate di acciaio e 60000 ore di lavoro. Inoltre, se l'azienda decide di produrre un tipo di veicolo, allora ne deve produrre almeno 1000 unità al mese.

Trovare il piano di produzione mensile che massimizza il profitto totale.

Problema 3 - formulazione

Variabili decisionali:

x_1 = numero di utilitarie prodotte, x_2 = numero di berline prodotte

x_3 = numero di station-wagon prodotte

$$y_1 = \begin{cases} 1 & \text{se l'azienda produce utilitarie,} \\ 0 & \text{altrimenti.} \end{cases} \qquad y_2 = \begin{cases} 1 & \text{se l'azienda produce berline,} \\ 0 & \text{altrimenti.} \end{cases}$$

$$y_3 = \begin{cases} 1 & \text{se l'azienda produce station-wagon,} \\ 0 & \text{altrimenti.} \end{cases}$$

Problema 3 - formulazione

Variabili decisionali:

x_1 = numero di utilitarie prodotte, x_2 = numero di berline prodotte

x_3 = numero di station-wagon prodotte

$$y_1 = \begin{cases} 1 & \text{se l'azienda produce utilitarie,} \\ 0 & \text{altrimenti.} \end{cases} \qquad y_2 = \begin{cases} 1 & \text{se l'azienda produce berline,} \\ 0 & \text{altrimenti.} \end{cases}$$

$$y_3 = \begin{cases} 1 & \text{se l'azienda produce station-wagon,} \\ 0 & \text{altrimenti.} \end{cases}$$

Funzione obiettivo: $2000\,x_1 + 3000\,x_2 + 4000\,x_3$ (da massimizzare)

Problema 3 - formulazione

Variabili decisionali:

x_1 = numero di utilitarie prodotte, x_2 = numero di berline prodotte

x_3 = numero di station-wagon prodotte

$$y_1 = \begin{cases} 1 & \text{se l'azienda produce utilitarie,} \\ 0 & \text{altrimenti.} \end{cases} \qquad y_2 = \begin{cases} 1 & \text{se l'azienda produce berline,} \\ 0 & \text{altrimenti.} \end{cases}$$

$$y_3 = \begin{cases} 1 & \text{se l'azienda produce station-wagon,} \\ 0 & \text{altrimenti.} \end{cases}$$

Funzione obiettivo: $2000\,x_1 + 3000\,x_2 + 4000\,x_3$ (da massimizzare)

Vincoli da rispettare:

$1.5\,x_1 + 3\,x_2 + 5\,x_3 \leq 6000$ (tonnellate di acciaio)

$30\,x_1 + 25\,x_2 + 40\,x_3 \leq 60000$ (ore di lavoro)

$$\left.\begin{array}{l} x_1 \leq 2000 y_1 \\ 1000 y_1 \leq x_1 \\ x_2 \leq 2000\, y_2 \\ 1000 y_2 \leq x_2 \\ x_3 \leq 1200\, y_3 \\ 1000 y_3 \leq x_3 \end{array}\right\} \quad \text{(vincoli di alternativa)}$$

$x_1, x_2, x_3 \geq 0, \quad x_1, x_2, x_3 \in \mathbb{Z}$

$y_1, y_2, y_3 \in \{0, 1\}$

Problema 3 - formulazione

Modello matematico (di programmazione lineare intera)

$$\begin{cases} \max\ 2000\,x_1 + 3000\,x_2 + 4000\,x_3 \\ 1.5\,x_1 + 3\,x_2 + 5\,x_3 \leq 6000 \\ 30\,x_1 + 25\,x_2 + 40\,x_3 \leq 60000 \\ x_1 \leq 2000 y_1 \\ 1000 y_1 \leq x_1 \\ x_2 \leq 2000\,y_2 \\ 1000 y_2 \leq x_2 \\ x_3 \leq 1200\,y_3 \\ 1000 y_3 \leq x_3 \\ x_1, x_2, x_3 \geq 0 \\ x_1, x_2, x_3 \in \mathbb{Z} \\ y_1, y_2, y_3 \in \{0, 1\} \end{cases}$$

Esercizio. Scrivere nella sintassi di AMPL il modello ed i dati relativi al problema precedente e trovare una soluzione ottima.

Problema 4 - descrizione

Un'azienda produce compressori d'aria e vuole pianificare la produzione ed i livelli di magazzino per i prossimi sei mesi. A causa delle fluttuazioni stagionali dei costi dei servizi e delle materie prime, il costo unitario di produzione dei compressori varia da un mese all'altro, così come la domanda di compressori. Anche la capacità produttiva varia da un mese all'altro a causa delle differenze nel numero di giorni lavorativi, delle ferie, della manutenzione programmata e della formazione del personale.
La tabella seguente riassume i costi di produzione mensili, la domanda e la capacità produttiva massima che la direzione dell'azienda prevede di dover affrontare nei prossimi sei mesi. Inoltre, per mantenere una forza lavoro stabile, l'azienda vuole produrre ogni mese almeno la metà della sua capacità produttiva massima.

	Mese					
	1	2	3	4	5	6
Costo produzione unitario (€)	240	250	265	285	280	260
Domanda	1000	4500	6000	5500	3500	4000
Produzione massima	4000	3500	4000	4500	4000	3500
Produzione minima	2000	1750	2000	2250	2000	1750

Problema 4 - descrizione

Date le dimensioni del magazzino, alla fine di ogni mese è possibile tenere in magazzino al massimo 6000 unità. L'azienda desidera tenere sempre in magazzino almeno 1500 unità come scorta di sicurezza per far fronte agli imprevisti della domanda. L'azienda stima che il costo di magazzino di un'unità in un dato mese sia pari all'1.5% del costo di produzione dell'unità nello stesso mese. L'azienda stima il numero di unità presenti in magazzino ogni mese calcolando la media delle scorte iniziali e finali di ogni mese. Attualmente ci sono 2750 unità in magazzino.

L'azienda vuole determinare il piano di produzione e di magazzino per i prossimi sei mesi, in modo da soddisfare la domanda prevista, con l'obiettivo di minimizzare i costi di produzione e di magazzino.

Problema 4 - formulazione

Variabili decisionali:
x_i = numero di unità prodotte nel mese i, con $i = 1, \dots, 6$.
y_i = numero di unità presenti in magazzino all'inizio del mese i, con $i = 1, \dots, 7$
(la situazione del magazzino alla fine del mese i è la stessa del magazzino all'inizio del mese $i + 1$)

Problema 4 - formulazione

Variabili decisionali:
$x_i =$ numero di unità prodotte nel mese i, con $i = 1, \dots, 6$.
$y_i =$ numero di unità presenti in magazzino all'inizio del mese i, con $i = 1, \dots, 7$
(la situazione del magazzino alla fine del mese i è la stessa del magazzino all'inizio del mese $i + 1$)

Funzione obiettivo: (da minimizzare)
$240x_1 + 250x_2 + 265x_3 + 285x_4 + 280x_5 + 260x_6$ (costo produzione)
$+3.6(y_1 + y_2)/2 + 3.75(y_2 + y_3)/2 + 3.98(y_3 + y_4)/2 + 4.28(y_4 + y_5)/2 + 4.20(y_5 + y_6)/2 + 3.9(y_6 + y_7)/2$ (costo magazzino)

Problema 4 - formulazione

Vincoli da rispettare:

$2000 \leq x_1 \leq 4000$ (produzione mese 1)
$1750 \leq x_2 \leq 3500$ (produzione mese 2)
$2000 \leq x_3 \leq 4000$ (produzione mese 3)
$2250 \leq x_4 \leq 4500$ (produzione mese 4)
$2000 \leq x_5 \leq 4000$ (produzione mese 5)
$1750 \leq x_6 \leq 3500$ (produzione mese 6)

$y_1 = 2750$ (livello magazzino all'inizio del mese 1)
$y_2 = y_1 + x_1 - 1000$ (livello magazzino all'inizio del mese 2)
$y_3 = y_2 + x_2 - 4500$ (livello magazzino all'inizio del mese 3)
$y_4 = y_3 + x_3 - 6000$ (livello magazzino all'inizio del mese 4)
$y_5 = y_4 + x_4 - 5500$ (livello magazzino all'inizio del mese 5)
$y_6 = y_5 + x_5 - 3500$ (livello magazzino all'inizio del mese 6)
$y_7 = y_6 + x_6 - 4000$ (livello magazzino all'inizio del mese 7)

$1500 \leq y_i \leq 6000$ per ogni $i = 2, \ldots, 7$ (capacità del magazzino)

Problema 4 - formulazione

Modello matematico (di programmazione lineare intera)

$$
\begin{cases}
\max \; 240x_1 + 250x_2 + 265x_3 + 285x_4 + 280x_5 + 260x_6 \\
\qquad +3.6(y_1 + y_2)/2 + 3.75(y_2 + y_3)/2 + 3.98(y_3 + y_4)/2 \\
\qquad +4.28(y_4 + y_5)/2 + 4.20(y_5 + y_6)/2 + 3.9(y_6 + y_7)/2 \\
2000 \leq x_1 \leq 4000 \\
1750 \leq x_2 \leq 3500 \\
2000 \leq x_3 \leq 4000 \\
2250 \leq x_4 \leq 4500 \\
2000 \leq x_5 \leq 4000 \\
1750 \leq x_6 \leq 3500 \\
y_1 = 2750 \\
y_2 = y_1 + x_1 - 1000 \\
y_3 = y_2 + x_2 - 4500 \\
y_4 = y_3 + x_3 - 6000 \\
y_5 = y_4 + x_4 - 5500 \\
y_6 = y_5 + x_5 - 3500 \\
y_7 = y_6 + x_6 - 4000 \\
1500 \leq y_i \leq 6000 \qquad \forall\, i = 2, \ldots, 7 \\
x_1, \ldots, x_6 \geq 0, \quad x_1, \ldots, x_6 \in \mathbb{Z} \\
y_1, \ldots, y_7 \geq 0, \quad y_1, \ldots, y_7 \in \mathbb{Z}
\end{cases}
$$

Problema 4 - formulazione AMPL

prod4.mod

```
#----------- parametri --------------------

param n integer > 0 ; # numero di mesi di pianificazione
param CostoProd{i in 1..n} >= 0 ;
# CostoProd[i] = costo di produzione unitario nel mese i

param CostoMag >= 0 ;
# costo unitario di magazzino = CostoMag * costo unitario di produzione

param domanda{i in 1..n} >= 0; # domanda[i] = domanda del mese i
param ProdMax{i in 1..n} >= 0 ;
# ProdMax[i] = produzione massima nel mese i

param FattoreProdMin >= 0 ;
# produzione minima nel mese i = FattoreProdMin * produzione massima nel mese i

param MagInizio >= 0; # livello iniziale del magazzino
param MagMin >= 0; # livello minimo del magazzino
param MagMax >= 0; # capacità del magazzino
```

Problema 4 - formulazione AMPL

prod4.mod

```
#------------- variabili -------------------

var x{i in 1..n} >= 0 ; # x[i] = numero di unità prodotte nel mese i
var y{i in 1..n+1} >= 0 ;
# y[i] = numero di unità presenti in magazzino all'inizio del mese i

#------------- funzione obiettivo ------------------

minimize CostoTot: sum{i in 1..n} (CostoProd[i]*x[i]
+ CostoMag*CostoProd[i]*(y[i]+y[i+1])/2 ) ;

#------------- vincoli -------------------

s.t. vin_MaxProd{i in 1..n}: x[i] <= ProdMax[i] ;
s.t. vin_MinProd{i in 1..n}: x[i] >= FattoreProdMin*ProdMax[i] ;

s.t. vin_LivMag1: y[1] = MagInizio ;
s.t. vin_LivMag{i in 2..n+1}: y[i] = y[i-1] + x[i-1] - domanda[i-1];

s.t. vin_MinMag{i in 2..n+1}: y[i] >= MagMin ;
s.t. vin_MaxMag{i in 2..n+1}: y[i] <= MagMax ;
```

Problema 4 - formulazione AMPL

prod4.dat

```
param n := 6;
param CostoProd :=
1 240
2 250
3 265
4 285
5 280
6 260;
param CostoMag := 0.015;
param domanda :=
1 1000
2 4500
3 6000
4 5500
5 3500
6 4000;
param ProdMax :=
1 4000
2 3500
3 4000
4 4500
5 4000
6 3500;
param FattoreProdMin := 0.5;
param MagInizio := 2750; param MagMin := 1500; param MagMax := 6000;
```

Problema 1 - descrizione

Un cliente desidera investire 750000€ nelle seguenti obbligazioni:

Compagnia	Rendimento annuo	Anni alla scadenza	Rating
Acme Chemical	8.65%	11	1 (eccellente)
DynaStar	9.50%	10	3 (buono)
Eagle Vision	10.00%	6	4 (discreto)
Micro Modeling	8.75%	10	1 (eccellente)
OptiPro	9.25%	7	3 (buono)
Sabre Systems	9.00%	13	2 (molto buono)

Il cliente ha le seguenti richieste:
- non vuole investire più del 25% della somma totale in una singola compagnia
- vuole che almeno il 50% della somma totale sia investito in obbligazioni a lungo termine (con almeno 10 anni alla scadenza)
- non vuole investire più del 35% della somma totale in compagnie con rating pari a 3 o 4.

Trovare un piano di investimento, che rispetti le richieste del cliente, in modo da massimizzare il rendimento totale annuo.

Problema 1 - formulazione

Variabili decisionali:

x_1 = euro da investire in Acme Chemical
x_2 = euro da investire in DynaStar
x_3 = euro da investire in Eagle Vision
x_4 = euro da investire in MicroModeling
x_5 = euro da investire in OptiPro
x_6 = euro da investire in Sabre Systems

Problema 1 - formulazione

Variabili decisionali:

x_1 = euro da investire in Acme Chemical
x_2 = euro da investire in DynaStar
x_3 = euro da investire in Eagle Vision
x_4 = euro da investire in MicroModeling
x_5 = euro da investire in OptiPro
x_6 = euro da investire in Sabre Systems

Funzione obiettivo: $0.0865x_1 + 0.095x_2 + 0.1x_3 + 0.0875x_4 + 0.0925x_5 + 0.09x_6$ (da massimizzare)

Problema 1 - formulazione

Variabili decisionali:

x_1 = euro da investire in Acme Chemical
x_2 = euro da investire in DynaStar
x_3 = euro da investire in Eagle Vision
x_4 = euro da investire in MicroModeling
x_5 = euro da investire in OptiPro
x_6 = euro da investire in Sabre Systems

Funzione obiettivo: $0.0865x_1 + 0.095x_2 + 0.1x_3 + 0.0875x_4 + 0.0925x_5 + 0.09x_6$ (da massimizzare)

Vincoli:

$x_1 + x_2 + x_3 + x_4 + x_5 + x_6 = 750000$ (cifra totale da investire)
$x_i \leq 187500$ per ogni $i = 1, \ldots, 6$ (max 25% investito in ogni compagnia)
$x_1 + x_2 + x_4 + x_6 \geq 375000$ (50% investito in obbligazioni a lungo termine)
$x_2 + x_3 + x_5 \leq 262500$ (max 35% investito in compagnie con rating 3 o 4)
$x_i \geq 0$ per ogni $i = 1, \ldots, 6$

Problema 1 - formulazione

Modello matematico (di programmazione lineare)

$$
\begin{cases}
\max \; 0.0865x_1 + 0.095x_2 + 0.1x_3 + 0.0875x_4 + 0.0925x_5 + 0.09x_6 \\
x_1 + x_2 + x_3 + x_4 + x_5 + x_6 = 750000 \\
x_i \leq 187500 \qquad \forall \; i = 1, \dots, 6 \\
x_1 + x_2 + x_4 + x_6 \geq 375000 \\
x_2 + x_3 + x_5 \leq 262500 \\
x_i \geq 0 \qquad \forall \; i = 1, \dots, 6
\end{cases}
$$

Problema 1 - formulazione AMPL

inv1.mod

```
#------------ parametri ---------------------
param budget >= 0; # somma da investire
param n integer > 0; # numero di obbligazioni
param rendimento{i in 1..n} > 0;
# rendimento[i] = rendimento percentuale dell'obbligazione i
param scadenza{i in 1..n} > 0;
# scadenza[i] = anni alla scadenza dell'obbligazione i
param rating{i in 1..n} > 0;
# rating[i] = rating dell'obbligazione i

param MaxPercSingola > 0;
# non investire più di MaxPercSingola% su una singola compagnia
param PercLungoTerm > 0;
# investire almeno PercLongTerm% in obbligazioni a lungo termine
#(scadenza >= 10 anni)
param PercBassoRating > 0;
# non investire più di PercBassoRating% in compagnie con rating 3 or 4
```

Problema 1 - formulazione AMPL

inv1.mod

```
#------------- variabili -------------------
var x{i in 1..n} >= 0; # x[i] = euro da investire nell'obbligazione i

#------------- funzione obiettivo -------------------
maximize RendimentoTotale: sum{i in 1..n} rendimento[i]*x[i]/100;

#------------- vincoli -------------------
s.t. v_budget: sum{i in 1..n} x[i] = budget;

s.t. v_MaxPercSingola{i in 1..n}: x[i] <= MaxPercSingola*budget/100;

s.t. v_PercLungoTerm: sum{i in 1..n: scadenza[i] >= 10}
 x[i] >= PercLungoTerm*budget/100;

s.t. v_PercBassoRating: sum{i in 1..n: rating[i] >= 3}
 x[i] <= PercBassoRating*budget/100;
```

Problema 1 - formulazione AMPL

inv1.dat

```
param budget := 750000;
param n := 6;
param rendimento :=
1 8.65
2 9.50
3 10.00
4 8.75
5 9.25
6 9.00;
param scadenza :=
1 11
2 10
3 6
4 10
5 7
6 13;
param rating :=
1 1
2 3
3 4
4 1
5 3
6 2;
param MaxPercSingola := 25;
param PercLungoTerm := 50;
param PercBassoRating := 35;
```

Problema 2 - descrizione

Supponiamo di voler investire un capitale di 100 mila euro.
Abbiamo a disposizione 9 investimenti possibili e per ciascuno di essi conosciamo il ricavo atteso ed il costo attuale:

	Investimento								
	1	2	3	4	5	6	7	8	9
Ricavo atteso (k€)	50	65	35	16	18	45	45	40	25
Costo (k€)	40	50	25	10	10	40	35	30	20

Compatibilmente con il capitale disponibile, quali investimenti dobbiamo fare in modo da massimizzare il ricavo totale atteso?

Problema 2 - formulazione

Variabili decisionali:

$$x_i = \begin{cases} 1 & \text{se l'investimento } i \text{ viene scelto,} \\ 0 & \text{altrimenti,} \end{cases} \quad \text{per ogni } i = 1, \dots, 9$$

Problema 2 - formulazione

Variabili decisionali:

$$x_i = \begin{cases} 1 & \text{se l'investimento } i \text{ viene scelto,} \\ 0 & \text{altrimenti,} \end{cases} \quad \text{per ogni } i = 1, \ldots, 9$$

Funzione obiettivo:

$50x_1 + 65x_2 + 35x_3 + 16x_4 + 18x_5 + 45x_6 + 45x_7 + 40x_8 + 25x_9$ (da massimizzare)

Problema 2 - formulazione

Variabili decisionali:

$$x_i = \begin{cases} 1 & \text{se l'investimento } i \text{ viene scelto,} \\ 0 & \text{altrimenti,} \end{cases} \quad \text{per ogni } i = 1, \dots, 9$$

Funzione obiettivo:

$50x_1 + 65x_2 + 35x_3 + 16x_4 + 18x_5 + 45x_6 + 45x_7 + 40x_8 + 25x_9$ (da massimizzare)

Vincoli:

$40x_1 + 50x_2 + 25x_3 + 10x_4 + 10x_5 + 40x_6 + 35x_7 + 30x_8 + 20x_9 \leq 100$
(capitale a disposizione)

Problema 2 - formulazione

Modello matematico (di programmazione lineare binaria)

$$
\begin{cases}
\max \; 50x_1 + 65x_2 + 35x_3 + 16x_4 + 18x_5 + 45x_6 + 45x_7 + 40x_8 + 25x_9 \\
40x_1 + 50x_2 + 25x_3 + 10x_4 + 10x_5 + 40x_6 + 35x_7 + 30x_8 + 20x_9 \leq 100 \\
x_j \in \{0, 1\}, \qquad j = 1, \dots, 9
\end{cases}
$$

Problema 2 - formulazione AMPL

inv2.mod

```
#------------ parametri ----------------------
param budget >= 0; # somma da investire
param n integer > 0; # numero di investimenti possibili
param ricavo{i in 1..n} > 0;
# ricavo[i] = ricavo atteso dell'investimento i
param costo{i in 1..n} > 0;
# costo[i] = costo attuale dell'investimento i

#------------ variabili ----------------------
var x{i in 1..n} binary;
#x[i]=1 se scelgo l'investimento i, 0 altrimenti

#------------- funzione obiettivo -------------------
maximize RicavoTotale: sum{i in 1..n} ricavo[i]*x[i];

#------------- vincoli -------------------
s.t. v_budget: sum{i in 1..n} costo[i]*x[i] <= budget;
```

Problema 2 - formulazione AMPL

inv2.dat

```
param budget := 100;
param n := 9;
param : ricavo costo :=
1 50 40
2 65 50
3 35 25
4 16 10
5 18 10
6 45 40
7 45 35
8 40 30
9 25 20;
```

Problema dello zaino

Il modello del problema 2 coincide con il modello del cosiddetto problema dello zaino (knapsack problem):
dati n oggetti di valore $v_1, \dots, v_n$ e peso $p_1, \dots, p_n$, ed un contenitore di capacità C, quali oggetti inserisco nel contenitore, rispettando la sua capacità, in modo da massimizzare il valore totale degli oggetti inseriti?

Variabili: $$x_j = \begin{cases} 1 & \text{se oggetto } j \text{ viene inserito,} \\ 0 & \text{altrimenti,} \end{cases} \qquad j = 1, \dots, n.$$

Modello:

$$\begin{cases} \max \displaystyle\sum_{j=1}^{n} v_j\, x_j \\ \displaystyle\sum_{j=1}^{n} p_j\, x_j \leq C \\ x_j \in \{0,1\}, \quad j = 1, \dots, n \end{cases}$$

Problema semplice da definire, ma difficile da risolvere (NP-hard).

Problema 3 - descrizione

Un'azienda ha bisogno di un fondo per pagare 800 mila dollari per la costruzione di un nuovo ristorante nei prossimi 6 mesi. I pagamenti di 250 mila dollari devono essere effettuati alla fine dei mesi 2 e 4, mentre il pagamento finale di 300 mila dollari deve essere effettuato alla fine del mese 6. Possono essere usati i seguenti investimenti:

Investimento	Disponibile nel mese	Mesi alla scadenza	Rendimento alla scadenza
A	1, 2, 3, 4, 5, 6	1	1.8%
B	1, 3, 5	2	3.5%
C	1, 4	3	5.8%
D	1	6	11.0%

Problema 3 - descrizione

Investment	Cash Inflow/Outflow at the Beginning of Month						
	1	2	3	4	5	6	7
A_1	-1	1.018					
B_1	-1	<——>	1.035				
C_1	-1	<——>	<——>	1.058			
D_1	-1	<——>	<——>	<——>	<——>	<——>	1.11
A_2		-1	1.018				
A_3			-1	1.018			
B_3			-1	<——>	1.035		
A_4				-1	1.018		
C_4				-1	<——>	<——>	1.058
A_5					-1	1.018	
B_5					-1	<——>	1.035
A_6						-1	1.018
Req'd Payments (in $1,000s)	$0	$0	$250	$0	$250	$0	$300

L'azienda deve determinare il piano di investimento che consenta di rispettare il calendario dei pagamenti richiesto, in modo da minimizzare la quantità di denaro depositata nel fondo all'inizio del primo mese.

Problema 3 - formulazione

Variabili decisionali:

x_{A1} = migliaia di dollari investiti nell'investimento A all'inizio del mese 1
x_{B1} = migliaia di dollari investiti nell'investimento B all'inizio del mese 1
x_{C1} = migliaia di dollari investiti nell'investimento C all'inizio del mese 1
x_{D1} = migliaia di dollari investiti nell'investimento D all'inizio del mese 1
x_{A2} = migliaia di dollari investiti nell'investimento A all'inizio del mese 2
x_{A3} = migliaia di dollari investiti nell'investimento A all'inizio del mese 3
x_{B3} = migliaia di dollari investiti nell'investimento B all'inizio del mese 3
x_{A4} = migliaia di dollari investiti nell'investimento A all'inizio del mese 4
x_{C4} = migliaia di dollari investiti nell'investimento C all'inizio del mese 4
x_{A5} = migliaia di dollari investiti nell'investimento A all'inizio del mese 5
x_{B5} = migliaia di dollari investiti nell'investimento B all'inizio del mese 5
x_{A6} = migliaia di dollari investiti nell'investimento A all'inizio del mese 6

Problema 3 - formulazione

Variabili decisionali:

x_{A1} = migliaia di dollari investiti nell'investimento A all'inizio del mese 1
x_{B1} = migliaia di dollari investiti nell'investimento B all'inizio del mese 1
x_{C1} = migliaia di dollari investiti nell'investimento C all'inizio del mese 1
x_{D1} = migliaia di dollari investiti nell'investimento D all'inizio del mese 1
x_{A2} = migliaia di dollari investiti nell'investimento A all'inizio del mese 2
x_{A3} = migliaia di dollari investiti nell'investimento A all'inizio del mese 3
x_{B3} = migliaia di dollari investiti nell'investimento B all'inizio del mese 3
x_{A4} = migliaia di dollari investiti nell'investimento A all'inizio del mese 4
x_{C4} = migliaia di dollari investiti nell'investimento C all'inizio del mese 4
x_{A5} = migliaia di dollari investiti nell'investimento A all'inizio del mese 5
x_{B5} = migliaia di dollari investiti nell'investimento B all'inizio del mese 5
x_{A6} = migliaia di dollari investiti nell'investimento A all'inizio del mese 6

Funzione obiettivo:

$x_{A1} + x_{B1} + x_{C1} + x_{D1}$ (da minimizzare)

Problema 3 - formulazione

Vincoli:

$$1.018x_{A1} = x_{A2} \qquad \text{(flusso di cassa inizio mese 2)}$$

$$1.035x_{B1} + 1.018x_{A2} = x_{A3} + x_{B3} + 250 \qquad \text{(flusso di cassa inizio mese 3)}$$

$$1.058x_{C1} + 1.018x_{A3} = x_{A4} + x_{C4} \qquad \text{(flusso di cassa inizio mese 4)}$$

$$1.035x_{B3} + 1.018x_{A4} = x_{A5} + x_{B5} + 250 \qquad \text{(flusso di cassa inizio mese 5)}$$

$$1.018x_{A5} = x_{A6} \qquad \text{(flusso di cassa inizio mese 6)}$$

$$1.11x_{D1} + 1.058x_{C4} + 1.035x_{B5} + 1.018x_{A6} = 300 \qquad \text{(flusso di cassa inizio mese 7)}$$

$$x \geq 0$$

Problema 3 - formulazione

Modello matematico (di programmazione lineare)

$$
\begin{cases}
\min \; x_{A1} + x_{B1} + x_{C1} + x_{D1} \\
1.018 x_{A1} = x_{A2} \\
1.035 x_{B1} + 1.018 x_{A2} = x_{A3} + x_{B3} + 250 \\
1.058 x_{C1} + 1.018 x_{A3} = x_{A4} + x_{C4} \\
1.035 x_{B3} + 1.018 x_{A4} = x_{A5} + x_{B5} + 250 \\
1.018 x_{A5} = x_{A6} \\
1.11 x_{D1} + 1.058 x_{C4} + 1.035 x_{B5} + 1.018 x_{A6} = 300 \\
x \geq 0
\end{cases}
$$

Problema 3 - formulazione AMPL

inv3.mod

```
#----------- parametri ---------------------

param m integer > 0; # numero di investimenti
param n integer > 0; # numero di mesi

param A{i in 1..m, j in 1..n} binary;
# A[i,j] = 1 se investimento i è disponibile all'inizio del mese j

param scadenza{i in 1..m} > 0;
# scadenza[i] = mesi alla scadenza dell'investimento i

param rendimento{i in 1..m} > 0;
# rendimento[i] = rendimento alla scadenza dell'investimento i

param pagamento{j in 2..n+1} >= 0;
# pagamento[j] = pagamento richiesto all'inizio del mese j
```

Problema 3 - formulazione AMPL

inv3.mod

```
#------------- variabili -------------------

var x{i in 1..m, j in 1..n: A[i,j]=1} >= 0;
# x[i,j] = migliaia di dollari investiti nell'investimento i
# all'inizio del mese j

#------------- funzione obiettivo -------------------

minimize InvestimentoIniziale: sum{i in 1..m: A[i,1]=1} x[i,1];

#------------- vincoli -------------------

s.t. FlussoCassa{j in 2..n+1}:
sum{i in 1..m: j-scadenza[i]>=1 and A[i,j-scadenza[i]]=1}
(1+rendimento[i])*x[i,j-scadenza[i]]
= sum{i in 1..m: j<=n and A[i,j]=1} x[i,j] + pagamento[j];
# all'inizio del mese j:
# ricavi dovuti ad investimenti passati
# = nuovi investimenti + pagamenti
```

Problema 3 - formulazione AMPL

inv3.dat

```
param m := 4;
param n := 6 ;
param A: 1 2 3 4 5 6 :=
1       1 1 1 1 1 1
2         1 0 1 0 1 0
3         1 0 0 1 0 0
4         1 0 0 0 0 0;
param scadenza rendimento :=
1 1 0.018
2 2 0.035
3 3 0.058
4 6 0.11;
param pagamento :=
2 0
3 250
4 0
5 250
6 0
7 300;
```

Problema di facility location - descrizione

Un'azienda deve decidere dove costruire un certo numero di magazzini in modo da rifornire i suoi 100 clienti. Nella figura i cerchi neri rappresentano le posizioni dei clienti, mentre i 20 cerchi rossi sono le posizioni in cui è possibile costruire un magazzino.



Ogni cliente vuole essere rifornito da un solo magazzino. I costi di costruzione dei magazzini ed i costi di trasporto da ogni possibile magazzino ad ogni cliente sono indicati nel file `FL.txt`.

L'azienda deve decidere quanti magazzini costruire in totale e dove costruirli, e come rifornire i clienti, ossia decidere da quale magazzino sarà servito ogni cliente, in modo da minimizzare il costo totale (dato dalla somma dei costi di costruzione e di trasporto).

Problema di facility location: formulazione

Parametri:

$C_1, \ldots, C_{20}$ = costi di costruzione dei magazzini nelle 20 posizioni possibili

T_{ij} = costo di trasporto per rifornire il cliente i dal magazzino costruito nella posizione j ($i = 1, \ldots, 100,\ j = 1, \ldots, 20$)

Variabili decisionali:

$$y_j = \begin{cases} 1 & \text{se l'azienda costruisce un magazzino nella posizione } j \\ 0 & \text{altrimenti} \end{cases}$$

per ogni $j = 1, \ldots, 20$

$$x_{ij} = \begin{cases} 1 & \text{se il cliente } i \text{ sarà rifornito dal magazzino costruito nella posizione } j \\ 0 & \text{altrimenti} \end{cases}$$

per ogni $i = 1, \ldots, 100$, $j = 1, \ldots, 20$

Problema di facility location: formulazione

Funzione obiettivo: $$\underbrace{\sum_{j=1}^{20} C_j y_j}_{\text{costi di installazione}} + \underbrace{\sum_{i=1}^{100} \sum_{j=1}^{20} T_{ij} x_{ij}}_{\text{costi di trasporto}} \quad \text{(da minimizzare)}$$

Vincoli:

$\sum_{j=1}^{20} x_{ij} = 1$ per ogni $i = 1, \dots, 100$ (ogni cliente deve essere rifornito da un solo magazzino)

$x_{ij} \leq y_j$ per ogni $i = 1, \dots, 100$, $j = 1, \dots, 20$
(se non viene costruito un magazzino nella posizione j, allora nessun cliente può essere rifornito da quel magazzino)

Problema di facility location: formulazione

Modello matematico (di programmazione lineare binaria)

$$\min \sum_{j=1}^{20} C_j y_j + \sum_{i=1}^{100} \sum_{j=1}^{20} T_{ij} x_{ij}$$

$$\sum_{j=1}^{20} x_{ij} = 1 \qquad \forall\ i = 1, \ldots, 100$$

$$x_{ij} \le y_j \qquad \forall\ i = 1, \ldots, 100,\ \forall\ j = 1, \ldots, 20$$

$$x_{ij} \in \{0, 1\} \qquad \forall\ i = 1, \ldots, 100,\ \forall\ j = 1, \ldots, 20$$

$$y_j \in \{0, 1\} \qquad \forall\ j = 1, \ldots, 20$$

Problema di facility location: formulazione AMPL

FL.mod

```
#------------ parametri ---------------------
param m integer > 0; # numero di clienti
param n integer > 0;
# numero di posizioni in cui è possibile costruire un magazzino
param C{j in 1..n} > 0;
# C[j] = costo per costruire un magazzino nella posizione j
param T{i in 1..m, j in 1..n} > 0;
# T[i,j] = costo di traporto per rifornire il cliente i
# dal magazzino costruito nella posizione j

#------------ variabili -------------------

var x{i in 1..m, j in 1..n} binary;
# x[i,j] = 1 se il cliente i sarà rifornito
# dal magazzino costruito nella posizione j

var y{j in 1..n} binary;
# y[j] = 1 se l'azienda costruisce un magazzino nella posizione j
```

Problema di facility location: formulazione AMPL

FL.mod

```
#------------- funzione obiettivo -------------------

minimize CostoTotale: sum{j in 1..n} C[j]*y[j]
 + sum{i in 1..m, j in 1..n} T[i,j]*x[i,j];

#------------- vincoli -------------------

s.t. v_clienti{i in 1..m}: sum{j in 1..n} x[i,j] = 1;
# ogni cliente deve essere rifornito da un solo magazzino

s.t. v_clienti_magazzini{i in 1..m, j in 1..n}: x[i,j] <= y[j];
# se non viene costruito un magazzino nella posizione j,
# allora nessun cliente può essere rifornito da quel magazzino
```

Distribuzione di merce - descrizione

Un'azienda produttrice di pasta dispone di n depositi, aventi ciascuno una capacità mensile pari a U_j scatole di pasta $(j = 1, \ldots, n)$, e m centri di distribuzione, ognuno dei quali richiede d_i scatole di pasta al mese $(i = 1, \ldots, m)$. Il rifornimento della pasta dai depositi ai centri di distribuzione è organizzato in modo che ogni centro deve essere rifornito da un unico deposito. Il deposito j ha un costo di spedizione c_j per ogni scatola inviata ai centri di distribuzione ed un costo fisso di gestione mensile f_j nel caso in cui rifornisca almeno un centro.

L'azienda deve stabilire come rifornire mensilmente i centri di distribuzione in modo da minimizzare il costo totale.

Distribuzione di merce - formulazione

Variabili decisionali: $x_{ij} = \begin{cases} 1 & \text{se il centro } i \text{ è servito dal deposito } j, \\ 0 & \text{altrimenti,} \end{cases}$

$$y_j = \begin{cases} 1 & \text{se il deposito } j \text{ rifornisce almeno un centro,} \\ 0 & \text{altrimenti.} \end{cases}$$

Funzione obiettivo: $\underbrace{\sum_{j=1}^{n} c_j \sum_{i=1}^{m} d_i x_{ij}}_{\text{costi di spedizione}} + \underbrace{\sum_{j=1}^{n} f_j y_j}_{\text{costi di gestione}}$ (da minimizzare)

Vincoli:

$\sum_{j=1}^{n} x_{ij} = 1$ per ogni $i = 1, \dots, m$
(ogni centro deve essere rifornito da un unico deposito)

$\sum_{i=1}^{m} d_i x_{ij} \le U_j y_j$ per ogni $j = 1, \dots, n$ (vincoli di capacità)

Distribuzione di merce - formulazione

Modello matematico (di programmazione lineare binaria)

$$\min \sum_{j=1}^{n} c_j \sum_{i=1}^{m} d_i x_{ij} + \sum_{j=1}^{n} f_j y_j$$

$$\sum_{j=1}^{n} x_{ij} = 1 \qquad \forall\ i = 1, \dots, m$$

$$\sum_{i=1}^{m} d_i x_{ij} \leq U_j y_j \qquad \forall\ j = 1, \dots, n$$

$$x_{ij} \in \{0, 1\} \qquad \forall\ i = 1, \dots, m,\ \forall\ j = 1, \dots, n$$

$$y_j \in \{0, 1\} \qquad \forall\ j = 1, \dots, n$$

Distribuzione di merce - formulazione AMPL

pasta.mod

```
#------------ parametri ------------------
param n > 0; # depositi
param m > 0; # centri distribuzione
param U{j in 1..n} >0;
# U[j] = capacità deposito j
param d{i in 1..m} >0;
# d[i] = richiesta del centro di distribuzione i
param c{j in 1..n} >0;
# c[j] = costo spedizione unitario del deposito j
param f{j in 1..n} >0;
# f[j] = costo fisso di gestione del deposito j

#------------ variabili ------------------
var x{i in 1..m, j in 1..n} binary;
# x[i,j] = 1 se il centro i è servito dal deposito j

var y{j in 1..n} binary;
# y[j] = 1 se il deposito j rifornisce almeno un centro di distribuzione
```

Distribuzione di merce - formulazione AMPL

pasta.mod

```
#------------ funzione obiettivo ------------------

minimize CostoTotale: sum{j in 1..n, i in 1..m} c[j]*d[i]*x[i,j]
+ sum{j in 1..n} f[j]*y[j];

#------------ vincoli ------------------

s.t. v_clienti{i in 1..m}: sum{j in 1..n} x[i,j] = 1;
# ogni centro deve essere rifornito da un unico deposito

s.t. v_capacita{j in 1..n}: sum{i in 1..m} d[i]*x[i,j] <= U[j]*y[j];
# vincoli di capacità
```